A Realistic Large-Signal Microwave PHEMT Transistors Model for SPICE

J.M. Zamanillo, H. Ingelmo, C. Perez-Vega and A. Mediavilla

University of Cantabria, Communications Engineering Department (DICOM), Av. de los Castros s/n. 39005, Santander, Spain, Phone +34-942-200887

Abstract **— A comprehensive large-signal HEMT model that provides a realistic description of measured characteristics over all operating regions for different PHEMTs is presented. The model was previously tested in harmonic-balance based simulators [1] and for the first time it has been implemented inside the time domain SPICE simulator. In order to do that, a new set of routines and libraries has been developed. The procedure introduced here can be extended to properly simulate other kind of devices described in terms of equivalent circuits. DC and scattering simulation results show very good agreement with the experimental measurements.**

I. Introduction

The potential of microwave MESFETs and PHEMTs active devices in actual communications systems has stimulated research efforts in our group with the development of different large-signal models for harmonic balance based simulators. These models offer results totally in concordance with the experimental measurements [1]-[4], DC and pulsed I-V curves, scattering parameters and CW laser optical stimulation of the device. Commercial simulators have the disadvantage of high cost, that makes them affordable only by certain manufactures and universities, because of the discount policy by educational licenses. Bearing in mind this idea, and in a trial to popularize the models developed by our group, emerges the following question: Why do not use SPICE as an microwave devices simulator?, this software is a universally known well tested tool. In the 70's of the past century, SPICE was developed by the University of Berkley, and nowadays several commercial simulators use the original code of SPICE for time domain simulations. This is one of the reasons that makes SPICE one of the most famous and popular electrical simulators used by the scientific community and electronic manufacturers. All the active device models included in PSPICE, including the GaAs MESFET, are based on physical characteristics [5] and therefore the parameters that define them depend on the transistor production technology used, as well as geometric factors, usually, known only by the manufacturer or the designer of the device. Unfortunately, SPICE lacks a built-in HEMT physical model, when SPICE has been used in the past to simulate GaAs MESFET and HEMT circuits, the standard JFET model was often employed in place of a true physical GaAs FET model. However, this approach has been shown [6]-[8] to yield considerable error in the various computed circuit responses. In this paper, we describe the implementation of our AlGaAs model of PHEMTs, previously reported [1]-[2] into PSPICE (commercial version for PC of the original Berkeley SPICE 2G code). This task has been divided into two parts: the first one, which is the aim of the present paper, describes how the microwave PHEMT model has been introduced in the SPICE simulator as well as the results obtained. The simulated characteristic I/V curves and scattering parameters have been validated with experimental measurements carried out in laboratory. In a second stage, which is now under development, and whose results will be reported in future communications, intermodulation effects including high order derivatives of the non-linear Ids current source must be simulated with the model.

II. The DC Model

Among the different existing classical models in the bibliography, our large-signal model [1]-[2] was chosen, because it is adequate to represent with high degree of fidelity the behavior of these devices, and furthermore the model is valid for HEMT and MESFET microwave transistors.

The DC large-signal equivalent circuit model proposed is shown in Fig. 1. Classical models offer, in general, good results for low-power transistors; however they are not capable of adequately represent power HEMT devices. The model implemented here is adequate for this kind of devices and is also capable of simulate the trasconductance compression phenomena in HEMT devices.

The expression for the current source Ids is given by (1) and it uses the internal voltages of the transistor Vgi and Vdi as variables, these voltages are shown in Fig.1. The physical meanings of the internal parameters used in this expression are shown in Table 1.



Fig. 1. DC Large Signal Model for PHEMT devices.

Parameter	Physical Meaning	S.I. units
Idss	Saturation drain current for Vgs=0 V.	**A**
V_{to}	Pinch-off Voltage	**V**
E	Parameter of the saturation current variation.	**-**
K_e	Parameter of the saturation current variation.	**V^{-1}**
S_s	Saturation drain current slope	**Ω^{-1}**
S_l	Drain current slope in the lineal zone for Vgi=0 V.	**Ω^{-1}**
K_g	Parameter that describes de lineal zone	**V^{-1}**
Vpf	Voltage at which transconductance degradation begins	**V**
δ	Transconductance adjustment parameter	-
μ	Transconductance adjustment parameter	**V^{δ}**

TABLE I

IDS CURRENT PARAMETERS

The relation between external and internal control voltages is shown in (2) and (3). The rest of the equations that compose the complete model has been omitted, since they have been previously reported and a detailed description of them can be found in [1]-[2]. It is very important to emphasize the accuracy in the extraction procedure of the parasitic resistance values because of their implication in the value of the intrinsic voltages.

$$Ids = Idss \cdot \left(1 - \frac{V_{gi}}{V_t}\right)^{(E+Ke \cdot Vg)} \cdot \left(1 + \frac{S_s \cdot V_{di}}{I_{dss}}\right) \cdot \tanh\left(\frac{Sl \cdot V_{di}}{1 - K_g \cdot V_{gi}}\right) \cdot e^{\left(-\frac{\left|V_{pf} - V_{gi}\right|^{\delta}}{\mu}\right)} \qquad (1)$$

Where $V_t = V_{to} + \gamma_{Ids} \cdot V_{di}$

$$V_{gi} = V_{gs} - I_d \cdot R_s \qquad (2)$$

$$V_{di} = V_{ds} - I_d \cdot \left(R_d + R_s\right) \qquad (3)$$

The model is valid for MESFET and HEMT microwave transistors. It must be noted this being a DC model, parasitic inductances Lg, Ld, Ls, as well as the parasitic capacitances Cpgi and Cpdi, have been omitted.

III. MODEL IMPLEMENTATION IN PSPICE

To guarantee the compatibility with all versions of the SPICE simulator, schematics had not been used in the description of the model components, and the old style commands from SPICE 2G written in readable ASCII text files have been used to integrate the model within SPICE. The same procedure has been used to implement all the libraries needed for a correct simulation. This work has been structured in three different tasks as follows:

- Task 1: Development of a Mathematical Library.
- Task 2: Definition of necessary subcircuits.
- Task 3: Building a main program that calls the above mentioned library and subcircuits.

```
************************************************************
*                LIBRERIA MATEMATICA : MATH.LIB           *
************************************************************
*-----------------------------------------------------------
* Descripcion de los SUBCKT Matematicos
*
* X1 1 2 3      ELEVA       V3=V1**V2
* X1 1 2 3      DIVIDE      V3=V1/V2
* X1 1 2 3      MULTIPLICA2 V3=V1*V2
* X1 1 2 3 4    MULTIPLICA3 V4=V1*V2*V3
* X1 1 2 3 4 5  MULTIPLICA4 V5=V1*V2*V3*V4
* X1 1 2 3      SUMA        V3=V1+V2
* X1 1 2 3      RESTA       V3=V1-V2
* X1 1 2 3      VTOI        Corriente de 2 a 3 = V1
* X1 1 2 3      VTOV        Tension de 2 a 3 = V1
* X1 1 2 3      GT          V3=1 si V1>V2
*                           V3=0 si V1<V2
* X1 1 2 3 4 5  COMP        V5=V1 si V3>V4
*                           V5=V2 si V3<V4
*-----------------------------------------------------------
* X1 1 2   KMUL    PARAMS:KKMUL= 2.0  V2=KKMUL*V1
* X1 1 2   P1      PARAMS:KAP1 KBP1  V2=A*V1+B
* X1 1 2   P2      PARAMS:KAP2 KBP2 KCP2 V2=A*V1^2+B*V1+C
* X1 1 2   ELEVAK  PARAMS:KELEVAK=2 V2=V1**KELEVAK
* X1 1 2   EXPONEN PARAMS:KEXPONEN=5 V2=EXP(KEXPONEN*V1)
* X1 1 2   LOGNEP  V2=LN(V1)
* X1 1 2   TANH    V2=TANH(V1)
* X1 1 2   COSH    V2=COSH(V1)
* X1 1 2   HFON    PARAMS:TAU=2.E-6 V2=V1*1/(1+S.Tau)
* X1 1 2   JFON    PARAMS:TAU=2.E-6 JFON=1-HFON
* X1 1 2   ABS     V2=ABS(V1)
*-----------------------------------------------------------
```

Fig. 2. Description of the mathematical functions contained in the library MATH.LIB (*the subcircuits have been omitted here because its extension*).

The first step necessary is to develop a mathematical library called *MATH.LIB* which contains a series of sub-circuits that implement typical mathematical functions (addition, subtraction, multiplication, hyperbolic tangent, exponentiation, etc.).

Ids current source	Node Implementation
	• **Nodes 1 & 2: INPUT voltage Vgi** • **Nodes 3 & 4: INPUT voltage Vdi** • **Nodes 5 & 6: OUTPUT current** **The linear capacitance Cds is not included in the sub-circuit.**

TABLE II

EXPLANATION OF IDS CURRENT INTERNAL NODES USED IN SUB-CIRCUIT HEMT_IDS

The second task consists, in the definition of the necessary sub-circuits that compose the complete model, one for each non linear current source (Ids, Igs and Igd), as well as the necessary components to properly model the non-linear capacitance Cgs, and the linear capacitance Cgd.

```
********************************************
*   LIBRERIAS: TRANSISTOR HEMT             *
********************************************
.SUBCKT HEMTIDS 1 2 3 4 5 6 PARAMS: KIDSS=81.952e-3
+ KVTO=-0.4223 GAMMAIDS=-0.075295 KE=1.3471 KKE=-0.4397
+ SS=0.003386 SL=2.3711 KG=-1.0811 KVPF=1.1723
+ DELTA=0.07497 + MU=1.1585
*Corriente en 5 y 6 como funcion de tension entre 1,2 Vgi
*y 3,4 Vdi
*IDS=IDSS*(1-Vgi/Vt)^(E+KE*Vgi)* (1+SS*Vdi/IDSS)*
*TANH(SL*Vdi/1-Kg*Vgi)*EXP(-|Vpf-Vgi|^DELTA/MU)
R1 1 0 1E12
R2 2 0 1E12
EDIFF1 102 0 VALUE={V(1)-V(2)}
RDIFF1 102 0 1
R3 3 0 1E12
R4 4 0 1E12
EDIFF2 304 0 VALUE={V(3)-V(4)}
RDIFF2 304 0 1
EVT 305 0 VALUE={KVTO + GAMMAIDS*V(304)}
RVT 305 0 1
XDIV1 102 305 103 DIVIDE
VDC1 104 0 DC 1
RDC1 104 0 1
XRES1 104 103 105 RESTA
E1 106 0 VALUE={KE+KKE*V(102)}
*XP1 102 106 P1      PARAMS: KAP1={KKE} KBP1={KE}
XPOT1 105 106 107 ELEVA
XKMUL1 107 108 KMUL    PARAMS: KKMUL= {KIDSS}
E2 306 0 VALUE={SS*V(304)/KIDSS}
XSUM1 306 104 307 SUMA
XKMUL2 304 308 KMUL    PARAMS: KKMUL= {SL}
XKMUL3 102 109 KMUL    PARAMS: KKMUL= {KG}
XRES2 104 109 110 RESTA
XDIV2 308 110 309 DIVIDE
XTANH 309 310 TANH
E4 111 0 VALUE={KVPF-V(102)}
XABS 111 112 ABS
*E5 404 0 VALUE={PWR(V(403),DELTA)/MU}
XELEV 112 113 ELEVAK  PARAMS: KELEVAK= {DELTA}
VDC2 114 0 {MU}
RDC2 114 0 1
XDIV3 113 114 115 DIVIDE
XEXP 115 116 EXPONEN PARAMS: KEXPONEN=-1.0
XMULT 108 307 310 116 400 MULTIPLICA4
XOUT 400 5 6 VTOI
.ENDS HEMTIDS
```

Fig 3. Description of HEMT_IDS library.

```
TRANSISTOR HEMT - (H3S) DIMENSIONES:4X30um

****************************************************
*   CIRCUITO PARA LA SIMULACION TRANSISTOR HEMT    *
****************************************************
*************
* LIBRERIAS *
*************
.LIB C:/PSPICE/MATH.LIB
.LIB C:/PSPICE/HEMT_GSD.LIB
******************************
*  DESCRIPCION DEL CIRCUITO  *
******************************
V1 1 0 DC 0.75
LG 1 2A 1E-10
R1 2A 2 6.6
XD1 2 5 DIODO_HEMT_GS PARAMS: INS=2E-13 ALPHA=14 CJO=0.1P
+ VBI=0.7 +M=0.5
XD2 2 4 DIODO_HEMT_GD PARAMS: INS=2E-15 ALPHA=13 CJO=0.1P
+ VBI=0.7 +M=0.5
V2 3 0 DC 6
LD 3 4A 1E-10
R2 4A 4 6.1
X1 2 5 4 5 4 5 HEMTIDS
CDS 4 5 15f
R3 5 5A 7.1
LS 5A 0 1E-10
**************************
*  ANALISIS EN CONTINUA  *
**************************
.DC V2 0 6 .5 V1 -0.7 1 0.1
***************
*  PROBE+END  *
***************
.PROBE
.END
```

Fig 4. Main program used to simulate the microwave PHEMT Transistor in PSPICE.

To illustrate this, Table 2 shows how the non-linear current source has been implemented in PSPICE using equation (1). The source code that describes the behavior of Ids has been named as: *HEMT_IDS.LIB*. It has six nodes, four input nodes and two output nodes, as shown in table II. The library *HEMT_GSD.LIB* is very similar to *HEMT_IDS.LIB*, and contains the sub-circuits related to the sources Igs and Igd, as well as the capacitances Cgs and Cgd.

Finally, a main program that calls the libraries and sub-circuits, written in SPICE language is needed to run the simulation. The content of this file is shown in Fig. 4. This procedure can be generalized to simulate any type of electronic device that can be characterized by an equivalent electrical circuit. With the necessary equations for the correct operation of the circuit proposed its integration in the simulator PSPICE student's version of ORCAD can be performed.



Fig 5. Simulation results of OMMIC PHEMTs in PSPICE. (a) Simulation of 4x30µm PHEMT in PSPICE. Ids vs. Vds. (b) Simulation of 4x30µm PHEMT in PSPICE. Ids vs. Vgs. (c) Comparison between simulated and experimental measurements of characteristics I-V curves for a 6*150µm (6 fingers, and 50µm of gate width) PHEMT device.

IV. RESULTS

In order to test the behavior of the model implemented in PSPICE simulator, two different size PHEMT devices from OMMIC foundry have been used. The first transistor is a low power device, 4*30µm (4 fingers, and 30µm of gate periphery) and the other is a medium power transistor 6*150µm (6 fingers, and 150µm of gate periphery). Fig. 2 (a) and Fig. 2 (b) show the PSPICE output by means of the graphic post-processor PROBE for the 4*30µm device.

Because SPICE is unable to import and plot external measurement data files, a small program written in MATLAB language shows a comparison between the SPICE results read directly of the output file .OUT and the experimental DC measurements performed in laboratory for the 6*150µm, transistor, taking the Vgs



(a)



(b)

Fig 6. Simulation results of OMMIC PHEMTs in PSPICE.
(a) Comparison between simulated and experimental measurements of characteristics I-V curves for a 4*30µm (4 fingers, and 30µm of gate perifery) PHEMT device.
(b) Comparison between simulated and experimental measurements of characteristics I-V curves for a 6*150µm (6 fingers, and 150µm of gate perifery) PHEMT device.

and Vds sweeps in the same experimentally measured values.

V. Conclusion

An electrical large-signal model for PHEMT and MESFET devices has been implemented in the PSPICE simulator, extending its utilization to this type of microwave devices. The method presented here can be extended to other types of non-linear devices using the same procedure. Simulation results show good experimental agreement, being similar to those of harmonic balance type simulators, including the transconductance compression phenomena. So it may be concluded that using a free tool as is the student version PSPICE, accurate results for microwave devices can be obtained. As future work lines, we continue with the development of new routines that allow to get the higher order derivatives for this type of devices, and the introduction of the electro-optical MESFET/HEMT transistors model [9,11] in the same way the model here presented has been introduced.

Acknowledgement

This work has been partially supported by the Spanish Regional Government of Cantabria through a R&D Sodercan contract and the *Spanish Comisión Interministerial de Ciencia y Tecnología* (CICYT) through the project CICYT-TIC2002-04084-C03-03. Furthermore, authors would to thank the project NoE TARGET (IST program of the EU under contract IST-1-707893-NOE) the facilities for the realization of the present work.

References

[1] J.M. Zamanillo Sainz de la Maza. "Metodología para la Extración Lineal y No Lineal de modelos Circuitales para Dispositivos MESFET y HEMT de media/alta Potencia. *PhD. Thesis. University of Cantabria*, Santander, April 1996.

[2] A. Mediavilla, A Tazón, J.L. García, T. Fernandez, J.A. García, J. M. García, C. Navarro, J. M. Zamanillo. "Dynamic Properties and Modelling of Large Signal, Termal, Optical and Intermodulation Effects in Microwave GaAs Devices", *Invited paper on the IEEE MTT-S* Workshop, Vol. 1, Denver (USA). June 8-13 1997.

[3] J.M. Zamanillo, C. Navarro, C. Pérez-Vega, A. Mediavilla, and A Tazón "Large Signal Model Predicts Dynamic Behavior MESFET Under Optical Illumination". *Microwave and Optical Technology Letters*. Vol. 29, No. 1, pp 25-31. April 5 2001.

[4] C. Navarro, J.M. Zamanillo, A. Mediavilla, A. Tazón, , J.L. García, M. Lomer and J.M. López-Higuera. "An Accurate Photonic Capacitance Model for GaAs MESFET". *IEEE Transactions on MTT*, Vol. 50, No. 4, pp 1193-1197, April 2002.

[5] G. Massobrio, P. Antognetti *Semiconductor device Modelling with SPICE*, Mc. Graw Hill 1993.

[6] W. R. Curtice, "A MESFET model for use in the design of GaAs integrated circuits," *IEEE Trans. Microwave Theory Tech* , Vol MTT-28, pp. 448-456, May 1980.

[7] S.E. Sussman-Fort, J.C. Hantgan and F.L. Huang. "A SPICE Model for Enhancement-and Depletion-Mode GaAs FET's.", *Transactions on Microwave Theory and Techniques* 34.11.Nov. 1986, pp. 1115-1119.

[8] A.J. McCamant, G.D. McCormack and D.H. Smith. "An Improved GaAs MESFET Model for SPICE (Short Papers)." *Transactions on Microwave Theory and Techniques* 38.6, Jun. 1990, pp 822-824.

[9] J. M. Zamanillo, C. Navarro, C. Pérez-Vega, J. A. García, A. Mediavilla and A. Tazón. "New Large Signal Model of AlGaAs PHEMT and GaAs MESFET Under Optical Illumination" *GaAs 2002 European Gallium Arsenide, Applications Symposium Proceedings*, pp. 121-124. Milan, Italy. Sept. 2002

[10] J.M. Zamanillo, J. Portilla, C. Navarro, C. Pérez-Vega and A. Mediavilla, "Optical Control of a GaAs MMIC Amplifier at S Band". *GaAs 2003 European Gallium Arsenide, Applications Symposium*. pp.117-120, .Munich, Germany. Oct. 2003.

[11] J.M. Zamanillo, C. Navarro, J. Sáiz-Ipiña, C. Perez-Vega and A. Mediavilla "New Large Signal Electrical Model of GaAs MESFET Under Optical Illumination". GaAs 2001 European Gallium Arsenide, Applications Symposium Procedings, pp.167-170, London, Sept-2001.